



Synth Nodes and UGen graphs

```
(
  SynthDef("simple", {
    var sig;
    sig = Saw.ar(200);
    sig = RLPF.ar(sig, 500, 0.1);
    Out.ar(0, sig);
  }).play;
)
```




math operators will become UGens

```
(
  SynthDef("mod", {
    var sig, resfreq;
    sig = Saw.ar(200);
    resfreq = SinOsc.kr(0.7) * 200 ;
    sig = RLPF.ar(sig, 500 + resfreq, 0.1);
    Out.ar(0, sig);
  }).play;
)
```




multi-channel expansion

```
(
  SynthDef("modstereo", {
    var sig, resfreq;
    sig = Saw.ar([200, 230]);
    resfreq = SinOsc.kr(0.7) * 200 ;
    sig = RLPF.ar(sig, 500 + resfreq, 0.1);
    Out.ar(0, sig);
  }).play;
)
```




control and audio buses

```
(
  SynthDef("bustest", {
    var sig, res;
    res = In.kr(0) + 100;
    sig = In.ar(0);
    sig = RLPF.ar(sig, res, 0.1);
    ReplaceOut.ar(0, sig);
  }).play;
)
```



function arguments will become Controls


```
(
  SynthDef("argtest", {
    arg freq=100, res=820;
    var sig;
    sig = Saw.ar(freq);
    sig = RLPF.ar(sig, res, 0.1);
    Out.ar(0, sig);
  }).play;
)
```



static random numbers

```
(
  SynthDef("rrandtest", {
    var sig;
    sig = Saw.ar(rrand(100, 300));
    Out.ar(0, sig);
  }).store;
)
```


```
Synth("rrandtest");
Synth("rrandtest");
```



scalar or init rate UGens

```
(
  SynthDef("randfreq", {
    var sig;
    sig = Saw.ar(Rand(100, 300));
    Out.ar(0, sig);
  }).store;
)
```

```
Synth("randfreq");
Synth("randfreq");
```



methods and UnaryOpUGens

```
(
  SynthDef("sineroot", {
    var sig;
    sig = SinOsc.ar(300, 0, 0.5.sqrt).sqrt;
    Out.ar(0, sig);
  }).play;
)
```